

## ЛОГИЧЕСКИЕ ФУНКЦИИ В EXCEL

Логические функции в Excel проверяют данные и возвращают результат «ИСТИНА», если условие выполняется, и «ЛОЖЬ», если не выполняется. Рассмотрим синтаксис логических функций и примеры применения их в процессе работы с программой Excel.

Функция ЕСЛИ возвращает один результат в случаях, если аргумент «Логическое выражение» при вычислении получает ответ «ИСТИНА» и другой, если ответ «ЛОЖЬ». Функция может содержать до семи логических условий внутри себя для аргумента «значение если истина», а для аргумент «значение если ложь» умеет выполнять функции или команды дальнейшего действия (к примеру, использование функции ПЕРЕЙТИ).


### Синтаксис функции:

=ЕСЛИ(\_лог\_выражение\_,\_значение\_если\_истина\_,\_значение\_если\_ложь\_)

- Логическое выражение – указывается практически любое выражение или значение, которое в процессе вычисления получает значение «ИСТИНА» или «ЛОЖЬ»;
- Значение если истина – это значение, которое будет возвращено, если аргумент «Логическое выражение» получит результат «ИСТИНА». В случае, когда этот аргумент не указан в формуле, возвращается результат «ИСТИНА»;
- Значение если ложь — это значение, которое будет возвращено, если аргумент «Логическое выражение» получит результат «ЛОЖЬ». В случае, когда этот аргумент опущен в формуле, возвращается результат «ЛОЖЬ».

### СИНТАКСИС ФУНКЦИИ ЕСЛИ С ОДНИМ УСЛОВИЕМ

Пример 1:




The screenshot shows the Excel interface. The formula bar at the top displays the formula: `=ЕСЛИ(A1>20;"больше 20";"меньше или равно 20")`. Below the formula bar, a table is visible with columns A through I and rows 1 through 2. Cell A1 contains the value 35. Cell B1 contains the text "больше 20". A red arrow points to cell A1.

	A	B	C	D	E	F	G	H	I
1	35	больше 20							
2									

Оператор проверяет ячейку A1 и сравнивает ее с 20. Когда содержимое графы больше 20, появляется истинная надпись «больше 20». Нет – «меньше или равно 20». Слова в формуле необходимо брать в кавычки, так как это текстовые значения.

Пример 2. Чтобы получить допуск к экзамену, студенты группы должны успешно сдать зачет. Результаты занесем в таблицу с графами: список студентов, зачет, экзамен.



The screenshot shows the Excel interface. The formula bar at the top displays the formula: `=ЕСЛИ(B2="зач."; "допущен"; "не допущен")`. Below the formula bar, a table is visible with columns A through E and rows 1 through 7. The table contains student names, their scores (зач. or незач.), and whether they are allowed to take the exam (допущен or не допущен). A red arrow points to the bottom of the table.

	A	B	C	D	E
1	Фамилия имя студента	Зачет	Экзамен		
2	Иванов Д.	зач.	допущен		
3	Петров В.	зач.	допущен		
4	Сидоров П.	незач.	не допущен		
5	Павлюченко И.	зач.	допущен		
6	Семашко В.	незач.	не допущен		
7					

Оператор ЕСЛИ должен проверить не цифровой тип данных, а текстовый. Поэтому прописали в формуле B2= «зач.»». В кавычки берем, чтобы программа правильно распознала текст.

*Пример 3.* Вложение логических функций.

C2		=ЕСЛИ(B2=5;"отлично";ЕСЛИ(B2=4;"хорошо";"удовлетворительно"))					
	A	B	C	D	E	F	G
1	Фамилия имя	Баллы	Результат				
2	Иванов Д.	5	отлично				
3	Петров В.	4	хорошо				
4	Сидоров П.	5	отлично				
5	Павлюченко И.	3	удовлетворительно				
6	Семашко В.	4	хорошо				

Таблица для анализа успеваемости. Ученик получил 5 баллов – «отлично». 4 – «хорошо». 3 – «удовлетворительно».

*Пример 4.* В этом примере добавлено третье условие, подразумевающее наличие в таблице успеваемости «двоек».

C2		=ЕСЛИ(B2=5;"отлично";ЕСЛИ(B2=4;"хорошо";ЕСЛИ(B2=3;"удовлетворительно";"неудовлетворительно")))									
	A	B	C	D	E	F	G	H	I	J	
1	Фамилия имя	Баллы	Результат								
2	Иванов Д.	5	отлично								
3	Петров В.	2	неудовлетворительно								
4	Сидоров П.	5	отлично								
5	Павлюченко И.	3	удовлетворительно								
6	Семашко В.	4	хорошо								

## РАСШИРЕНИЕ ФУНКЦИОНАЛА С ПОМОЩЬЮ ОПЕРАТОРОВ «И», «ИЛИ»

Когда нужно проверить несколько истинных условий, используется функция И.

Функция ИЛИ проверяет условие 1 или условие 2. Как только хотя бы одно условие истинно, то результат будет истинным.

Функции И и ИЛИ могут проверить до 30 условий.

*Пример использования оператора И:*

D1		=ЕСЛИ(И(A1=B1;B1=C1;A1=C1);"значения равны";"значения не равны")							
	A	B	C	D	E	F	G	H	I
1	53	45	53	значения не равны					
2	43	45	46	значения не равны					
3	34	34	34	значения равны					

*Пример использования оператора ИЛИ:*

D1		=ЕСЛИ(ИЛИ(A1=B1;B1=C1;A1=C1);"есть равные значения";"нет равных значений")							
	A	B	C	D	E	F	G	H	I
1	53	45	53	есть равные значения					
2	43	45	46	нет равных значений					
3	34	34	34	есть равные значения					

*Пример 5.* Необходимо переоценить товарные остатки. Если продукт хранится на складе дольше 8 месяцев, уменьшить его цену в 2 раза. Сформируем таблицу с исходными параметрами:

	А	В	С
1	Название товара	Стоимость на момент поступления на склад	Срок хранения, мес.
2	Товар 1	200р.	6
3	Товар 2	500р.	12
4	Товар 3	700р.	5
5	Товар 4	400р.	7
6	Товар 5	600р.	10
7	Товар 6	300р.	3

Чтобы решить поставленную задачу, воспользуемся логической функцией ЕСЛИ. Формула будет выглядеть так: =ЕСЛИ(С2>=8;В2/2;В2).

Логическое выражение «С2>=8» построено с помощью операторов отношения «>» и «=». Результат его вычисления – логическая величина «ИСТИНА» или «ЛОЖЬ». В первом случае функция возвращает значение «В2/2». Во втором – «В2».

fx =ЕСЛИ(С2>=8;В2/2;В2)		
В	С	Д
Стоимость на момент поступления на склад	Срок хранения, мес.	Стоимость после переоценки
200р.	6	200р.
500р.	12	250р.
700р.	5	700р.
400р.	7	400р.
600р.	10	300р.
300р.	3	300р.

Усложним задачу – задействуем логическую функцию И. Теперь условие такое: если товар хранится дольше 8 месяцев, то его стоимость уменьшается в 2 раза. Если дольше 5 месяцев, но меньше 8 – в 1,5 раза.

fx =ЕСЛИ(И(С2>=8);В2/2;ЕСЛИ(И(С2>=5);В2/1,5;В2))		
В	С	Д
Стоимость на момент поступления на склад	Срок хранения, мес.	Стоимость после переоценки
200р.	6	133р.
500р.	12	250р.
700р.	5	467р.
400р.	7	267р.
600р.	10	300р.
300р.	3	300р.

*Пример 6.* Если стоимость товара на складе после уценки стала меньше 300 р. или продукт хранится дольше 10 месяцев, его списывают.

Для решения используем логические функции ЕСЛИ и ИЛИ:

=ЕСЛИ(ИЛИ(D2<300;C2>=10);"списан"; "")

. Условие, записанное с помощью логической операции ИЛИ, расшифровывается так: товар списывается, если число в ячейке D2 = 10.

=ЕСЛИ(ИЛИ(D2<300;C2>=10);"списан"; "")

С	D	E
Срок хранения, мес.	Стоимость после переоценки	
6	133р.	списан
12	250р.	списан
5	467р.	
7	267р.	списан
10	300р.	списан
3	300р.	

При невыполнении условия функция ЕСЛИ возвращает пустую ячейку.

В качестве аргументов можно использовать другие функции. К примеру, математические.

*Пример 7.* Ученики перед поступлением в гимназию сдают математику, русский и английский языки. Проходной балл – 12. По математике для поступления нужно получить не менее 4 баллов. Составить отчет о поступлении.

Нужно общее количество баллов сравнить с проходным баллом. И проверить, чтобы по математике оценка была не ниже «4». В графе «Результат» поставить «принят» или «нет».

Введем формулу вида:

=ЕСЛИ(И(B3>=4;СУММ(B3:D3)>=\$B\$1);"принят";"нет")

. Логический оператор «И» заставляет функцию проверять истинность двух условий. Математическая функция «СУММ» используется для подсчета итогового балла.

=ЕСЛИ(И(B3>=4;СУММ(B3:D3)>=\$B\$1);"принят";"нет")

	A	B	C	D	E
1	Проходной балл	12			
2	Фамилия, имя	Математика	Русск.яз.	Англ.яз.	Результат
3	Иванов В.	3	5	4	нет
4	Петров П.	4	4	4	принят
5	Сидоров И.	4	3	5	принят
6	Катюшкина Е.	4	3	3	нет
7	Симонова А.	5	3	4	принят

*Пример 8.* Проанализировать стоимость товарных остатков после уценки. Если цена продукта после переоценки ниже средних значений, то списать со склада этот продукт.

	A	B	C
1	Название товара	Стоимость на момент поступления на склад	Срок хранения, мес.
2	Товар 1	200р.	6
3	Товар 2	500р.	12
4	Товар 3	700р.	5
5	Товар 4	400р.	7
6	Товар 5	600р.	10
7	Товар 6	300р.	3

Для решения задачи используем формулу вида:

=ЕСЛИ(D2<СРЗНАЧ(D2:D7);"списан";"")

В логическом выражении «D2<СРЗНАЧ(D2:D7)» применена статистическая функция СРЗНАЧ. Она возвращает среднее арифметическое в диапазоне D2:D7.

=ЕСЛИ(D2<СРЗНАЧ(D2:D7);"списан";"")		
C	D	E
Срок хранения, мес.	Стоимость после переоценки	
6	133р.	списан
12	250р.	списан
5	467р.	
7	267р.	списан
10	300р.	
3	300р.	

*Пример 9.* Найти средние продажи в магазинах сети.

	A	B	C
1	Товар	Магазин	Продажи
2	Товар 1	№1	234876
3	Товар 2	№2	346589
4	Товар 3	№3	543890
5	Товар 4	№1	275684
6	Товар 5	№2	549650
7	Товар 6	№3	178560

Необходимо найти среднее арифметическое для ячеек, значение которых отвечает заданному условию. То есть совместить логическое и статистическое решение.

Ниже таблицы с условием составим табличку для отображения результатов:

8			
9	Среднее	№1	
10	Среднее	№2	
11	Среднее	№3	

Решим задачу с помощью функции:

=СРЗНАЧЕСЛИ(\$B\$2:\$B\$7;B9;\$C\$2:\$C\$7)

. Первый аргумент – \$B\$2:\$B\$7 – диапазон ячеек для проверки. Второй аргумент – B9 – условие. Третий аргумент – \$C\$2:\$C\$7 – диапазон усреднения; числовые значения, которые берутся для расчета среднего арифметического.

Функция СРЗНАЧЕСЛИ сопоставляет значение ячейки B9 (№1) со значениями в диапазоне B2:B7 (номера магазинов в таблице продаж). Для совпадающих данных считает среднее арифметическое, используя числа из диапазона C2:C7.

fx		=СРЗНАЧЕСЛИ(\$B\$2:\$B\$7;B9;\$C\$2:\$C\$7)			
	A	B	C	D	E
1	Товар	Магазин	Продажи		
2	Товар 1	№1	234876		
3	Товар 2	№2	346589		
4	Товар 3	№3	543890		
5	Товар 4	№1	275684		
6	Товар 5	№2	549650		
7	Товар 6	№3	178560		
8					
9	Среднее	№1	255280		
10	Среднее	№2	448119,5		
11	Среднее	№3	361225		

*Пример 10.* Найти средние продажи в магазине №1 г. Москва.

Видоизменим таблицу из предыдущего примера:

	A	B	C	D
1	Товар	Магазин	Город	Продажи
2	Товар 1	№1	Москва	234876
3	Товар 2	№2	Москва	346589
4	Товар 3	№3	Новосибирск	543890
5	Товар 4	№1	Новосибирск	275684
6	Товар 5	№1	Москва	549650
7	Товар 6	№3	Новосибирск	178560
8				
9	Среднее	№1	Москва	

Нужно выполнить два условия – воспользуемся функцией вида:

=СРЗНАЧЕСЛИМН(\$D\$2:\$D\$7;\$B\$2:\$B\$7;B9;\$C\$2:\$C\$7;C9)

fx		=СРЗНАЧЕСЛИМН(\$D\$2:\$D\$7;\$B\$2:\$B\$7;B9;\$C\$2:\$C\$7;C9)				
	A	B	C	D	E	F
1	Товар	Магазин	Город	Продажи		
2	Товар 1	№1	Москва	234876		
3	Товар 2	№2	Москва	346589		
4	Товар 3	№3	Новосибирск	543890		
5	Товар 4	№1	Новосибирск	275684		
6	Товар 5	№1	Москва	549650		
7	Товар 6	№3	Новосибирск	178560		
8						
9	Среднее	№1	Москва	392263		

Функция СРЗНАЧЕСЛИМН позволяет применять более одного условия. Первый аргумент – \$D\$2:\$D\$7 – диапазон усреднения (откуда берутся цифры для нахождения среднего

арифметического). Второй аргумент – \$B\$2:\$B\$7 – диапазон для проверки первого условия. Третий аргумент – B9 – первое условие. Четвертый и пятый аргумент – диапазон для проверки и второе условие, соответственно.

Функция учитывает только те значения, которые соответствуют всем заданным условиям.

## ФУНКЦИЯ СЧЕТЕСЛИ

Функция СЧЕТЕСЛИ входит в группу статистических функций. Позволяет найти число ячеек по определенному критерию. Работает с числовыми, текстовыми значениями, датами.

Рассмотрим аргументы функции:

- Диапазон – группа значений для анализа и подсчета (обязательный).
- Критерий – условие, по которому нужно подсчитать ячейки (обязательный).

В диапазоне ячеек могут находиться текстовые, числовые значения, даты, массивы, ссылки на числа. Пустые ячейки функция игнорирует.

В качестве критерия может быть ссылка, число, текстовая строка, выражение. Функция СЧЕТЕСЛИ работает только с одним условием (по умолчанию). Но можно ее «заставить» проанализировать 2 критерия одновременно.

Рекомендации для правильной работы функции:

- Если функция СЧЕТЕСЛИ ссылается на диапазон в другой книге, то необходимо, чтобы эта книга была открыта.
- Аргумент «Критерий» нужно заключать в кавычки (кроме ссылок).
- Функция не учитывает регистр текстовых значений.
- При формулировании условия подсчета можно использовать подстановочные знаки. «?» - любой символ. «\*» - любая последовательность символов. Чтобы формула искала непосредственно эти знаки, ставим перед ними знак тильды (~).
- Для нормального функционирования формулы в ячейках с текстовыми значениями не должно пробелов или непечатаемых знаков.

Посчитаем числовые значения в одном диапазоне. Условие подсчета – один критерий.

	А	В
1	Столы	124
2	Стулья	235
3	Табуреты	380
4	Кровати	100
5	Шкафы 1	59
6	Столы	87
7	Табуреты	210
8	Шкафы 2	241
9	Тумбы	386
10	Стулья	45
11	Банкетки	87

Посчитаем количество ячеек с числами больше 100. Формула: =СЧЁТЕСЛИ(B1:B11;">100"). Диапазон – B1:B11. Критерий подсчета – «>100». Результат:

fx =СЧЁТЕСЛИ(B1:B11;">100")		
D	E	F
>100	6	

Если условие подсчета внести в отдельную ячейку, можно в качестве критерия использовать ссылку:

fx =СЧЁТЕСЛИ(B1:B11;D2)		
D	E	F
>100	6	

Посчитаем текстовые значения в одном диапазоне. Условие поиска – один критерий.

Формула: =СЧЁТЕСЛИ(A1:A11;"табуреты"). Или:

fx =СЧЁТЕСЛИ(A1:A11;D3)		
D	E	F
>100	6	
табуреты	2	

Во втором случае в качестве критерия использовали ссылку на ячейку.

Формула с применением знака подстановки: =СЧЁТЕСЛИ(A1:A11;"таб\*").

Для расчета количества значений, оканчивающихся на «и», в которых содержится любое число знаков: =СЧЁТЕСЛИ(A1:A11;"\*и"). Получаем:

fx =СЧЁТЕСЛИ(A1:A11;"*и")		
D	E	F
>100	6	
табуреты	2	
*и	2	

Формула посчитала «кровати» и «банкетки».

Используем в функции СЧЕТЕСЛИ условие поиска «не равно».

Формула: =СЧЁТЕСЛИ(A1:A11;"<>"&"стулья"). Оператор «<>» означает «не равно». Знак амперсанда (&) объединяет данный оператор и значение «стулья».



fx =СЧЁТЕСЛИ(A1:A11;"<>"&"стулья")			
	D	E	F
>100		6	
табуреты		2	
*и		2	
<>стулья		9	

При применении ссылки формула будет выглядеть так:

fx =СЧЁТЕСЛИ(B1:B11;"<>"&D6)			
	D	E	F
>100		6	
табуреты		2	
*и		2	
<>стулья		9	
100		10	

Часто требуется выполнять функцию СЧЕТЕСЛИ в Excel по двум критериям. Таким способом можно существенно расширить ее возможности.

Рассмотрим специальные случаи применения СЧЕТЕСЛИ в Excel и примеры с двумя условиями. Функция СЧЕТЕСЛИ с двумя условиями в Excel очень часто используется для автоматизированной и эффективной работы с данными.

1. Посчитаем, сколько ячеек содержат текст «столы» и «стулья». Формула: =СЧЁТЕСЛИ(A1:A11;"столы")+СЧЁТЕСЛИ(A1:A11;"стулья"). Для указания нескольких условий используется несколько выражений СЧЕТЕСЛИ. Они объединены между собой оператором «+».

fx =СЧЁТЕСЛИ(A1:A11;"столы")+СЧЁТЕСЛИ(A1:A11;"стулья")						
	A	B	C	D	E	F
1	Стол	124				
2	Стулья	235		4		
3	Табуреты	380				
4	Кровати	100				
5	Шкафы 1	59				
6	Стол	87				
7	Табуреты	210				
8	Шкафы 2	241				
9	Тумбы	386				
10	Стулья	45				
11	Банкетки	87				

2. Условия – ссылки на ячейки. Формула:  
=СЧЁТЕСЛИ(A1:A11;A1)+СЧЁТЕСЛИ(A1:A11;A2). Текст «столы» функция ищет в ячейке A1. Текст «стулья» - на базе критерия в ячейке A2.

fx =СЧЁТЕСЛИ(A1:A11;A1)+СЧЁТЕСЛИ(A1:A11;A2)

	A	B	C	D	E
1	Стол	124			
2	Стул	235		4	
3	Табурет	380			
4	Кровать	100			
5	Шкаф 1	59			
6	Стол	87			
7	Табурет	210			
8	Шкаф 2	241			
9	Тумба	386			
10	Стул	45			
11	Банкетка	87			

3. Посчитаем число ячеек в диапазоне B1:B11 со значением большим или равным 100 и меньшим или равным 200. Формула: =СЧЁТЕСЛИ(B1:B11;">=100")-СЧЁТЕСЛИ(B1:B11;">200").

fx =СЧЁТЕСЛИ(B1:B11;">=100")-СЧЁТЕСЛИ(B1:B11;">200")

	A	B	C	D	E	F
1	Стол	124				
2	Стул	235		2		
3	Табурет	380				
4	Кровать	100				
5	Шкаф 1	59				
6	Стол	87				
7	Табурет	210				
8	Шкаф 2	241				
9	Тумба	386				
10	Стул	45				
11	Банкетка	87				

4. Применим в формуле СЧЁТЕСЛИ несколько диапазонов. Это возможно, если диапазоны являются смежными. Формула: =СЧЁТЕСЛИ(A1:B11;">=100")-СЧЁТЕСЛИ(A1:B11;">200"). Ищет значения по двум критериям сразу в двух столбцах. Если диапазоны несмежные, то применяется функция СЧЁТЕСЛИМН.

fx =СЧЁТЕСЛИ(A1:B11;">=100")-СЧЁТЕСЛИ(A1:B11;">200")

	A	B	C	D	E	F
1	Стол	124				
2	Стул	235		2		
3	Табурет	380				
4	Кровать	100				

5. Когда в качестве критерия указывается ссылка на диапазон ячеек с условиями, функция возвращает массив. Для ввода формулы нужно выделить такое количество ячеек, как в диапазоне с критериями. После введения аргументов нажать одновременно сочетание клавиш Shift + Ctrl + Enter. Excel распознает формулу массива.

fx {=СЧЁТЕСЛИ(A1:A11;D2:D4)}		
С	Д	Е
	Условия	Результат
	столы	2
	стулья	2
	табуреты	2

## ПРОМЕЖУТОЧНЫЕ.ИТОГИ И СЧЕТЕСЛИ

Посчитаем количество реализованных товаров по группам.

1. Сначала отсортируем таблицу так, чтобы одинаковые значения оказались рядом.

	А	В
1	Банкетки	87
2	Кровати	100
3	Стол	124
4	Стол	87
5	Стулья	235
6	Стулья	45
7	Табуреты	380
8	Табуреты	210
9	Тумбы	386
10	Шкафы 1	59
11	Шкафы 2	241

2. Первый аргумент формулы «ПРОМЕЖУТОЧНЫЕ.ИТОГИ» - «Номер функции». Это числа от 1 до 11, указывающие статистическую функцию для расчета промежуточного результата. Подсчет количества ячеек осуществляется под цифрой «2» (функция «СЧЕТ»).

fx =ПРОМЕЖУТОЧНЫЕ.ИТОГИ(2;A5:B6)				
	А	В	С	Д
1	Банкетки	87		
2	Кровати	100		2
3	Стол	124		
4	Стол	87		
5	Стулья	235		
6	Стулья	45		
7	Табуреты	380		
8	Табуреты	210		
9	Тумбы	386		
10	Шкафы 1	59		
11	Шкафы 2	241		

Формула нашла количество значений для группы «Стулья». При большом числе строк (больше тысячи) подобное сочетание функций может оказаться полезным.