

Функции для работы с датами в Excel

Для работы с датами в Excel в разделе с функциями определена категория «Дата и время». Рассмотрим наиболее распространенные функции в этой категории.

Программа Excel «воспринимает» дату и время как обычное число. Электронная таблица преобразует подобные данные, приравнивая сутки к единице. В результате значение времени представляет собой долю от единицы. К примеру, 12.00 – это 0,5.

Значение даты электронная таблица преобразует в число, равное количеству дней от 1 января 1900 года (так решили разработчики) до заданной даты. Например, при преобразовании даты 13.04.1987 получается число 31880. То есть от 1.01.1900 прошло 31 880 дней.

Этот принцип лежит в основе расчетов временных данных. Чтобы найти количество дней между двумя датами, достаточно от более позднего временного периода отнять более ранний.

Пример функции ДАТА

Построение значение даты, составляя его из отдельных элементов-чисел.

Синтаксис: год; месяц, день.

Все аргументы обязательные. Их можно задать числами или ссылками на ячейки с соответствующими числовыми данными: для года – от 1900 до 9999; для месяца – от 1 до 12; для дня – от 1 до 31.

Если для аргумента «День» задать большее число (чем количество дней в указанном месяце), то лишние дни перейдут на следующий месяц. Например, указав для декабря 32 дня, получим в результате 1 января.

Пример использования функции:

fx		=ДАТА(2003;7;26)	
	A	B	
1	26.07.2003		

Зададим большее количество дней для июня:

fx		=ДАТА(2004;6;31)	
	A	B	
1	01.07.2004		

Примеры использования в качестве аргументов ссылок на ячейки:

D2	fx		=ДАТА(A2;B2;C2)	
	A	B	C	D
1	Год	Месяц	День	Дата
2	0	2	1	01.02.1900
3	1	13	2	02.01.1902
4	1987	5	32	01.06.1987

Функция РАЗНДАТ в Excel

Возвращает разницу между двумя датами.

Аргументы:

- начальная дата;
- конечная дата;
- код, обозначающий единицы подсчета (дни, месяцы, годы и др.).

Способы измерения интервалов между заданными датами:

- для отображения результата в днях – «d»;
- в месяцах – «m»;
- в годах – «y»;
- в месяцах без учета лет – «ym»;
- в днях без учета месяцев и лет – «md»;
- в днях без учета лет – «yd».

В некоторых версиях Excel при использовании последних двух аргументов («md», «yd») функция может выдать ошибочное значение. Лучше применять альтернативные формулы.

Примеры действия функции РАЗНДАТ:

C5		fx		=РАЗНДАТ(A1;B1;C4)				
	A	B	C	D	E	F	G	H
1	28.02.2009	30.03.2012						
2								
3								
4			d	m	y	ym	md	yd
5			1126	37	3	1	2	30

Функция ГОД в Excel

Возвращает год как целое число (от 1900 до 9999), который соответствует заданной дате. В структуре функции только один аргумент – дата в числовом формате. Аргумент должен быть введен посредством функции ДАТА или представлять результат вычисления других формул.

Пример использования функции ГОД:

fx		=ГОД(A1)	
	A	B	
1	23.05.2015	2015	
2	17.01.1987	1987	

Функция МЕСЯЦ в Excel

Возвращает месяц как целое число (от 1 до 12) для заданной в числовом формате даты. Аргумент – дата месяца, который необходимо отобразить, в числовом формате. Даты в текстовом формате функция обрабатывает неправильно.

Примеры использования функции МЕСЯЦ:

fx =МЕСЯЦ(A1)		
	А	В
1	23 май 15	5
2	1987, 17 января	1

Примеры функций ДЕНЬ, ДЕНЬНЕД и НОМНЕДЕЛИ в Excel

Возвращает день как целое число (от 1 до 31) для заданной в числовом формате даты. Аргумент – дата дня, который нужно найти, в числовом формате.

fx =ДЕНЬ(A1)		
	А	В
1	23 май 15	23
2	1987, 17 января	17

Чтобы вернуть порядковый номер дня недели для указанной даты, можно применить функцию ДЕНЬНЕД. Функция ДЕНЬНЕД возвращает день недели, который соответствует определенной дате. Результат работы данной функции – это число из диапазона от 1 до 7 (единица соответствует воскресенью, а семь – субботе). По умолчанию функция считает воскресенье первым днем недели.

fx =ДЕНЬНЕД(A1)		
	А	В
1	24 май 15	1
2	1987, 17 января	7

Функция ДЕНЬНЕД имеет следующую синтаксическую запись:

=ДЕНЬНЕД(дата;[тип])

Описание аргументов функции:

- Дата – обязательный параметр, соответствующей дате, день недели которой требуется определить. Данный параметр имеет особенности, описанные в примечаниях к предыдущей функции.
- Тип – число в диапазоне от 1 до 3, указывающее на то, какой день недели считается первым (в некоторых странах первый день недели – воскресенье, в нашей – понедельник). Если тип=1, первым днем недели будет считаться воскресенье, тип=2 – понедельник. Если тип=3, результатом работы функции ДЕНЬНЕД будет число из диапазона от 0 до 6, где 0 соответствует понедельнику.

В некоторых случаях вместо чисел 1,2...7 требуется вывод названия дня недели (понедельник, вторник... воскресенье). Для решения данной задачи можно использовать следующую функцию: =ТЕКСТ(A2;"ДДДД").

	А	В
1	Исходная дата	День недели
2	17.06.2018	воскресенье
3	31.12.2020	четверг
4	20.02.2020	четверг
5	08.08.2008	пятница
6	12.12.2012	среда
7	08.03.2020	воскресенье
8	11.07.2020	суббота

Результатом выполнения данной функции является текст «воскресенье». Если требуется краткая запись дня недели (Сб, Пн, Ср), в качестве второго параметра следует указать «ДДД».

Для отображения порядкового номера недели для указанной даты применяется функция НОМНЕДЕЛИ:




fx		=НОМНЕДЕЛИ(A1)
	А	В
1	24 май 15	22

Дата 24.05.2015 приходится на 22 неделю в году. Неделя начинается с воскресенья (по умолчанию).

fx		=НОМНЕДЕЛИ(A1;2)
	А	В
1	24 май 15	21




В качестве второго аргумента указана цифра 2. Поэтому формула считает, что неделя начинается с понедельника (второй день недели).

Для указания текущей даты используется функция **СЕГОДНЯ** (не имеет аргументов). Чтобы отобразить текущее время и дату, применяется функция **ТДАТА** ().

B1		:				=СЕГОДНЯ()
	A	B	C	D		
1	Текущая дата:	06.09.2018	<--	=СЕГОДНЯ()		
2	Текущие дата и время:	06.09.2018 10:45	<--	=ТДАТА()		

Для удобочитаемого представления текущей даты функцию СЕГОДНЯ можно использовать в формуле вместе с текстом. Для этого ее лучше поместить в качестве первого аргумента для функции ТЕКСТ и указать соответствующий формат для отображения значения. Формула для отображения сегодняшней даты в формате день, месяц и год:

= "Сегодня "&ТЕКСТ(СЕГОДНЯ();"дддд д ММММ гггг")&"г."

B1		:				="Сегодня "&ТЕКСТ(СЕГОДНЯ());"дддд д ММММ гггг")&"г."					
	A	B	C	D	E	F					
1	Текущая дата:	Сегодня четверг 6 Сентябрь 2018г.									

Аналогичным способом можно отображать любую дату и время:

A3	:	X	✓	fx	=Валюта введена в оборот: "&ТЕКСТ(A2;"дддд д ММММ гггг")&"г."
	A	B	C	D	E
1	Дата введения банкнот и монет ЕВРО в Европейском союзе				
2	01.01.2002				
3	Валюта введена в оборот: вторник 1 Январь 2002г.				

Для этого нужно убрать из формулы функцию СЕГОДНЯ и вместо нее ввести статические значения вручную или указать ссылку на ячейку с датой и временем.

ФУНКЦИЯ ДАТАЗНАЧ ДЛЯ ПРЕОБРАЗОВАНИЯ ТЕКСТА В ДАТУ В EXCEL

Функция ДАТАЗНАЧ в Excel предназначена для работы с текстовыми данными в формате ДАТА. Она используется для преобразования текстовых данных в формат Дата и возвращает числовое значение, характеризующее указанную дату.

Функция ДАТАЗНАЧ возвращает число, соответствующее числовому представлению даты, которая указана в виде текста, с учетом указанной выше особенности хранения дат в Excel. Формат возвращаемого значения зависит от настроек формата ячейки, в которой будет выведен результат вычислений.

Зачастую даты в Excel записывают без использования функции ДАТА. Табличный редактор определяет такие значения как обычные текстовые строки. Поэтому процедуры форматирования, сортировки по дате, а также различные вычисления (например, разница дат) приводят к некорректным результатам или появлению ошибок. Поэтому функция ДАТАЗНАЧ полезна для преобразования текстовых значений к данным формата Дата.

Пример 1. В таблице Excel находится столбец, в котором хранятся даты как текстовые строки, при этом записи имеют вид: «28 сентября 2018 года». Преобразовать эти значения в данные формата Дата.

	A	B
1	Дата как текст	Дата
2	11 сентября 2001 года	
3	17 августа 2009 года	
4	08 апреля 2017 года	
5	23 июня 2018 года	
6	15 марта 2006 года	

Для получения даты в формате используем следующую функцию:

=ДАТАЗНАЧ(ЛЕВСИМВ(A2;2)&"-"&ПСТР(ЛЕВСИМВ(A2;6);4;3)&"-"&ПСТР(ПРАВСИМВ(A2;9);1;4))

Единственный аргумент состоит из подстрок, склеенных амперсандами (&):

- Функция ЛЕВСИМВ возвращает номер дня (первые два символа строки, содержащейся в ячейке A2). Очень важно, чтобы однозначные номера дней (например, 8 апреля) записывались как 08 апреля (имели ноль в начале), иначе будет возникать ошибка.
- Комбинация функций ПСТР и ЛЕВСИМВ выделяет из строки три первых символа названия месяца и возвращает их.
- Комбинация функций ПСТР и ПРАВСИМВ выделяет 4 символа, соответствующие числовому представлению года.

Растянем формулу вниз по столбцу, чтобы рассчитать остальные значения:

B2		✕ ✓ <i>fx</i>		=ДАТАЗНАЧ(ЛЕВСИМВ(A2;2)&"-"&ПСТР(ЛЕВСИМВ(A2;6);4;3)&"-"&ПСТР(ПРАВСИМВ(A2;9);1;4))		
	A	B	C	D	E	F
1	Дата как текст	Дата				
2	11 сентября 2001 года	11.09.2001				
3	17 августа 2009 года	17.08.2009				
4	08 апреля 2017 года	08.04.2017				
5	23 июня 2018 года	23.06.2018				
6	15 марта 2006 года	15.03.2006				

Таким образом мы преобразовали текстовые строки в формат даты, которые теперь можно использовать для вычислений в формулах.